

# Introdução a Programação

**Aula 05.2 – Estruturas de Seleção**

**Escolha-Caso**

Prof. Luiz Mário Lustosa Pascoal



# Portugol – Escolha-caso

**Comandos de testes de condições: *Escolha-Caso***

*É um comando que contém vários desvios*

- No ***Escolha-caso***, o computador testa uma variável sucessivamente contra uma lista de constantes inteiras ou de caracteres e executa um comando ou bloco de comandos quando encontrar uma coincidência.
- O comando ***Escolha-caso*** verifica o conteúdo de variáveis dos tipos: ***inteiro, caracter***.



# Sintaxe – Escolha-caso

```
escolha (variavel)  
{  
    caso constante1:  
        //comandos do caso constante1  
        pare // Impede que as instruções do caso 2  
              // sejam executadas  
    caso constante2:  
        //comandos do caso constante2  
        pare // Impede que as instruções do caso 3  
              // sejam executadas  
    caso constante3:  
        //comandos do caso constante3  
        pare  
    caso contrario:  
        // Será executado para qualquer opção diferente de 1, 2 ou 3  
}
```



# Portugol – Escolha-caso

## ***Dicas importantes:***

- *De modo geral, a sequência de comandos do **caso** não precisam estar entre chaves.*
- *Porém, se o primeiro comando dentro de um **caso** for a declaração de uma variável, será necessário colocar todos os comandos desse case dentro de um par de chaves.*



# Portugol – Escolha-caso

## pare

- Quando o computador encontra alguma coincidência, ele executa os comandos associados àquele **caso** até encontrar **pare** ou o fim do comando **escolha**.
- É um erro comum programadores esquecerem de colocar o **pare** após os comandos.



# Portugol – Escolha-caso

## *caso contrario*

- O comando **caso contrario** será executado se não for encontrada nenhuma coincidência na lista de constantes.
- Caso não seja colocado um comando **caso contrario** e não haja coincidência, nenhum comando será executado.



# Portugol – Escolha-caso

- **Exemplo:**

Escreva um programa que leia uma estação do ano e informe ao usuário qual o período que aquela estação ocorre.

- **Primavera (p):** Setembro a Dezembro.
- **Verão (v):** Dezembro a Março.
- **Outono (o):** Março a Junho.
- **Inverno (i):** Junho a Setembro



```
caracter estacao
```

```
escreva("P - Primavera\nV - Verão \nO - Outono \nI - Inverno\n")  
escreva("Informe qual estação deseja saber o período: ")  
leia(estacao)
```

```
escolha(estacao)
```

```
{  
    caso 'p':  
    caso 'P':  
        escreva("A Primavera ocorre de Setembro até Dezembro.")  
    pare  
    caso 'v':  
    caso 'V':  
        escreva("O Verão ocorre de Dezembro até Março.")  
    pare  
    caso 'o':  
    caso 'O':  
        escreva("O Outono ocorre de Março até Junho.")  
    pare  
    caso 'i':  
    caso 'I':  
        escreva("O Inverno ocorre de Junho até Setembro.")  
    pare  
    caso contrario:  
        escreva("Opção de estação inválida!")  
}
```





# Portugol – Escolha-caso

## *importante:*

- O **escolha** difere do **se**, já que o primeiro só pode testar igualdade e a expressão condicional **se** pode ser de qualquer tipo.
- Não pode haver duas constantes **caso** com valores iguais no mesmo **escolha**.
- Podem ser colocados outros comandos **escolha**, **se** dentro de comandos **escolha**.
- Pode ser deixado um **caso** vazio quando mais de uma condição usa o mesmo código.
- O **escolha-caso** é comumente utilizado para executar os comandos de um menu.



# Portugol – Escolha-caso

- **Exercício:**

Receba dois números reais do usuário em seguida exiba o resultado de uma operação escolhida também pelo usuário: Soma, Subtração, Multiplicação e Divisão.

- **Obs.:** No caso da divisão, se o segundo operando for zero exiba a mensagem “divisão impossível”.



```
real n1, n2
character operacao

escreva("Informe o primeiro numero: ")
leia(n1)
escreva("Informe o segundo numero: ")
leia(n2)

escreva("Informe a operação: (Soma (+); Subtração (-);",
        "Multiplicação (*); Divisão(/) ): ")
leia(operacao)

escolha (operacao)
{
    caso '+':
        escreva(n1, " + ", n2, " = ", n1+n2)
        pare
    caso '-':
        escreva(n1, " - ", n2, " = ", n1 - n2)
        pare
    caso '*':
        escreva(n1, " * ", n2, " = ", n1*n2)
        pare
    caso '/':
        se (n2 == 0)
            escreva("Divisão Impossível!")
        senao{
            escreva(n1, " / ", n2, " = ", n1/n2)
        }
        pare
    caso contrario:
        escreva("Operação inválida!")
}
```



# Portugol – Escolha-caso

- **Exercício 2:**
- Faça um programa que leia uma letra e mostre se ela é uma vogal ou consoante.
  - Suponha que o usuário só irá digitar letras.



**AGORA É  
PRATICAR!!!**

