

Estruturas de Repetição – Parte I

ENQUANTO-FAÇA

Prof. Dr. Edson Pimentel
Centro de Matemática, Computação e
Cognição



Objetivos

- Reconhecer a necessidade do uso de estruturas de repetição
- Aprender a **sintaxe** da Estrutura de Repetição **“ENQUANTO-FAÇA”** na “**linguagem PORTUGOL**”
- Aprender a resolver problemas que requeiram o uso de **ESTRUTURAS DE REPETIÇÃO** a partir de seus enunciados (com a aplicação da técnica de interpretação de enunciados)

Problema

(Estudo de Caso)

Problema

Escreva um algoritmo **calcular a média da idade** de um grupo de 10 pessoas.

- **SAÍDA:** Quais os resultados que o algoritmo dever fornecer ?
 - **ENTRADA:** Quais os insumos necessários para se obter a saída?
 - **PROCESSAMENTO:** Como transformar os insumos na saída?
-
- **SAÍDA:** média da idade das 10 pessoas
 - **ENTRADA:** a idade das 10 pessoas (**id1, id2, id3,..., id10**)
 - **PROCESSAMENTO:**
 - $Media = (id1 + id2 + id3 + \dots + id10) / \underline{10}$

Solução [1]

```
inteiro id1, id2, id3, id4, id5, id6, id7, id8, id9, id10  
real media
```

```
escreva ("Digite a idade da pessoa 1: ")  
leia(id1)  
  
escreva ("Digite a idade da pessoa 2: ")  
leia(id2)  
  
escreva ("Digite a idade da pessoa 3: ")  
leia(id3)  
■  
■  
■  
  
escreva ("Digite a idade da pessoa 9: ")  
leia(id9)  
  
escreva ("Digite a idade da pessoa 10: ")  
leia(id10)
```



**Entrada de
Dados**

Solução [1] - continuação



```
escreva ("Digite a idade da pessoa 10: ")  
leia(id10)
```



**Entrada de
Dados**

Processamento

```
media<- (id1+id2+id3+id4+id5+id6+id7+id8+id9+id10) / 10
```

Saída de Dados

```
escreva("A media das idades é: ", media)
```

**E se fossem 50 pessoas ?
100 pessoas ?
200 pessoas ?**



**50 variáveis ?
50 “escreva / leia” ?**

Padrão de Comportamento

```
inteiro id1, id2, id3, id4, id5, id6, id7, id8, id9, id10  
real media
```

```
escreva ("Digite a idade da pessoa 1: ")  
leia(id1)
```

```
escreva ("Digite a idade da pessoa 2: ")  
leia(id2)
```

```
escreva ("Digite a idade da pessoa 3: ")  
leia(id3)
```

```
escreva ("Digite a idade da pessoa 4: ")  
leia(id4)
```

```
escreva ("Digite a idade da pessoa 5: ")  
leia(id5)
```

```
escreva ("Digite a idade da pessoa 6: ")  
leia(id6)
```

Note que existe um
**“Padrão de
Comportamento”**
nesse algoritmo.

**Padrão de
Comportamento**



REPETIÇÃO




Universidade Federal do ABC

ESTRUTURAS DE REPETIÇÃO


Estruturas de Repetição

- Quando uma seqüência de comandos deve ser executada **repetidas vezes** usamos uma **estrutura de repetição**.
- A estrutura de repetição, assim como a de decisão, envolve sempre a avaliação de uma **condição**.
- Também conhecidas como **laços** de repetição ou **loops**.




Estrutura de Repetição: ENQUANTO

- A estrutura de repetição **enquanto** permite especificar instruções que devem ser repetida enquanto determinada condição for verdadeira
- **Exemplo**: enquanto não terminar o arquivo, leia linha e incremente contador de aprovados se for o caso.
- O corpo da estrutura **enquanto** pode ser uma instrução ou um bloco de execução



Estrutura de Repetição: ENQUANTO

- Quando a condição da estrutura **enquanto** se tornar **FALSA**, a ação (ou bloco de ações) da estrutura será pulada.
- O programa continuará com a ação imediatamente após a estrutura **enquanto**.




- IMPORTANTE:** Deve-se sempre prever o comando, ou ação, que tornará falsa a condição do comando **enquanto**, caso contrário, seu programa entrará em **loop infinito**.
- Condição de SAÍDA (do laço)**

Estrutura de Repetição: ENQUANTO

Pseudolinguagem

```
enquanto <condição> faça  
    <bloco_de_execução>  
  
fimenquanto
```



SINTAXE DO PORTUGOL

```
enquanto <condição> {  
    INSTRUÇÃO 1  
    ...  
}
```



Universidade Federal do ABC

O problema da Tabuada

Problema da Tabuada

**Escreva um algoritmo
para mostrar na tela a
TABUADA do número N.**

5	X	1	=	5
5	X	2	=	10
5	X	3	=	15
5	X	4	=	20
5	X	5	=	25
5	X	6	=	30
5	X	7	=	35
5	X	8	=	40
5	X	9	=	45
5	X	10	=	50

Problema da Tabuada

Escreva um algoritmo para mostrar na tela a TABUADA do número N.

- **SAÍDA:** mostrar na tela as 10 linhas com a tabuada do N
- **ENTRADA:** 1 número (n)
- **PROCESSAMENTO:**
 - $n * 1,$
 - $n * 2,$
 - $n * 3,$
 - ...
 - $n * 9,$
 - $n * 10$

```
inteiro n
```

```
escreva ("Digite um número : ")  
leia(n)
```

```
escreva ("\n", n, " X 1 = ", n * 1)  
escreva ("\n", n, " X 2 = ", n * 2)  
escreva ("\n", n, " X 3 = ", n * 3)  
escreva ("\n", n, " X 4 = ", n * 4)  
escreva ("\n", n, " X 5 = ", n * 5)  
escreva ("\n", n, " X 6 = ", n * 6)  
escreva ("\n", n, " X 7 = ", n * 7)  
escreva ("\n", n, " X 8 = ", n * 8)  
escreva ("\n", n, " X 9 = ", n * 9)  
escreva ("\n", n, " X 10 = ", n * 10)
```

Problema da Tabuada


Escreva um algoritmo para mostrar na tela a TABUADA do número N.

```
inteiro n

escreva ("Digite um número : ")
leia(n)

escreva ("\n", n, " X 1 = ", n * 1)
escreva ("\n", n, " X 2 = ", n * 2)
escreva ("\n", n, " X 3 = ", n * 3)
escreva ("\n", n, " X 4 = ", n * 4)
escreva ("\n", n, " X 5 = ", n * 5)
escreva ("\n", n, " X 6 = ", n * 6)
escreva ("\n", n, " X 7 = ", n * 7)
escreva ("\n", n, " X 8 = ", n * 8)
escreva ("\n", n, " X 9 = ", n * 9)
escreva ("\n", n, " X 10 = ", n * 10)
```

RESULTADO



```
Console  Mensagens
Digite um número : 5
5 X 1 = 5
5 X 2 = 10
5 X 3 = 15
5 X 4 = 20
5 X 5 = 25
5 X 6 = 30
5 X 7 = 35
5 X 8 = 40
5 X 9 = 45
5 X 10 = 50
```


Problema da Tabuada – Solução Alternativa

Escreva um algoritmo para mostrar na tela a TABUADA do número N.

```
inteiro n, c
```

```
escreva ("Digite um número : ")
```

```
leia(n)
```

```
c = 1
```

```
escreva ("\n", n, " X ", c, " = ", n * c)  
c = c + 1
```

```
escreva ("\n", n, " X ", c, " = ", n * c)  
c = c + 1
```

```
escreva ("\n", n, " X ", c, " = ", n * c)  
c = c + 1
```


```
// REPETIR MAIS 7 VEZES ...
```

C=1

C=2

C=3

REPETIÇÃO



Problema da Tabuada – Solução com repetição

Escreva um algoritmo para mostrar na tela a TABUADA do número N.

- **SAÍDA:** mostrar na tela as 10 linhas com a tabuada do N
- **ENTRADA:** 1 número (n)
- **PROCESSAMENTO:**
 - Imprimir $(n * y)$ 10 vezes

Problema da Tabuada – Solução com repetição

```
inteiro n, c
```

```
escreva ("Digite um número : ")
```

```
leia(n)
```

```
c = 1
```

C=1


```
enquanto (c <=10) {
```

```
    escreva ("\n", n, " X ", c, " = ", n * c)
```

```
    c = c + 1
```

```
}
```

C=1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11





Universidade Federal do ABC

Bacharelado em Ciência e Tecnologia Processamento da Informação

Estruturas de Repetição – Parte 1

IMPORTANTE !

Padrão de Instrução - CONTADOR

- Um “**padrão de instrução**” bastante utilizado em programas que requerem o uso de REPETIÇÃO é a figura do **CONTADOR**
- Um “contador” geralmente tem a seguinte forma
 - **var = var + 1** ou seja a variável “recebe” o valor que ela já possui somado com mais 1 ... e assim por diante ...

```
inteiro  x
x = 1
enquanto (x <= 5) (

    escreva(x)
    x = x + 1

}
```

O que esse trecho de programa faz ?

Padrão de Instrução - CONTADOR

```
inteiro  x
x = 1
enquanto (x <= 5) {

    escreva(x)
    x = x + 1
}
```

X
1
2
3
4
5
6

CONTADORES geralmente são inicializados com ZERO ou com 1

Padrão de Instrução - ACUMULADOR

- Um outro “**padrão de instrução**” bastante utilizado em programas que requerem o uso de REPETIÇÃO é a figura do **ACUMULADOR**
- Um “**ACUMULADOR**” geralmente tem a seguinte forma
 - **var = var + X** ou seja, a variável “recebe” o valor dela somado com mais X ... e assim por diante ...

```
inteiro  c, s, idade
c = 1
s = 0
enquanto (c <= 5) {
    escreva("Digite a idade ")
    leia (idade)
    s = s + idade // acumulador
    c = c + 1    // contador
}
escreva("s = ", s)
```

O que esse trecho de programa faz ?

Padrão de Instrução - ACUMULADOR

```
inteiro  c, s, idade
c = 1
s = 0
enquanto (c <= 5) {
    escreva("Digite a idade ")
    leia (idade)
    s = s + idade // acumulador
    c = c + 1     // contador
}
escreva("s = ", s)
```

C	S	IDADE
1	0	
		10
2	10	
		18
3	28	
		12
4	40	
		15
5	55	
		21
6	76	

ACUMULADORES geralmente são inicializados com o valor **ZERO**

Retomando ...

Média da idade de 50 pessoas

Estruturas de Repetição – Parte 1

Solução [1]

E se fossem 50 pessoas ?
100 pessoas ?
200 pessoas ?

```
inteiro id1, id2, id3, id4, id5, id6, id7, id8, id9, id10  
real media
```

```
escreva ("Digite a idade da pessoa 1: ")  
leia(id1)  
escreva ("Digite a idade da pessoa 2: ")  
leia(id2)  
escreva ("Digite a idade da pessoa 3: ")  
leia(id3)  
escreva ("Digite a idade da pessoa 4: ")  
leia(id4)  
escreva ("Digite a idade da pessoa 5: ")  
leia(id5)  
escreva ("Digite a idade da pessoa 6: ")  
leia(id6)  
escreva ("Digite a idade da pessoa 7: ")  
leia(id7)  
escreva ("Digite a idade da pessoa 8: ")  
leia(id8)  
escreva ("Digite a idade da pessoa 9: ")  
leia(id9)  
escreva ("Digite a idade da pessoa 10: ")  
leia(id10)  
media<-(id1+id2+id3+id4+id5+id6+id7+id8+id9+id10)/10  
escreva("A media das idades é: ", media)
```

Solução com REPETIÇÃO - ENQUANTO

```
inteiro cont = 1, soma = 0, idade  
real media
```

```
enquanto (cont <= 50) {
```

```
    escreva ("\nDigite a idade da pessoa ", cont, " : ")  
    leia(idade)
```

```
    soma = soma + idade
```

```
    cont = cont + 1
```


```
}
```

```
media = soma / 50
```

```
escreva("\nA media das idades é: ", media)
```

← Acumulador

← Contador





Outro Exemplo

Problema

Realizou-se uma pesquisa com 20 pessoas que responderam à seguinte pergunta: **Quantos filhos você tem?**

Escreva um algoritmo para processar essa pesquisa informando quantas pessoas possuem até 2 filhos e quantas possuem mais de 2 filhos

- **SAÍDA:** Quais os resultados que o algoritmo dever fornecer ?
 - **ENTRADA:** Quais os insumos necessários para se obter a saída?
 - **PROCESSAMENTO:** Como transformar os insumos na saída?
-
- **SAÍDA:** Quant. Pessoas até 2 filhos e Quant. Pessoas mais 2 filhos
 - **ENTRADA:** Quant. Filhos
 - **PROCESSAMENTO:**
 - Se quant \leq então
 acumular o contador 1
 - Senão
 acumular o contador 2

Problema: Quantos filhos você tem?

```
inteiro qfilhos, quant1, quant2, cont
```

```
quant1 = 0
```

```
quant2 = 0
```

```
cont = 0
```

```
enquanto (cont < 20) {
```

```
    escreva ("\nQuantos filhos ? ")
```

```
    leia(qfilhos)
```

```
    se (qfilhos <= 2) {
```

```
        quant1 = quant1 + 1
```

```
    }
```

```
    senao {
```

```
        quant2 = quant2 + 1
```

```
    }
```

```
    cont = cont + 1
```

```
}
```

Entrada de Dados

Processamento

Problema: Quantos filhos você tem?

```
inteiro qfilhos, quant1, quant2, cont
```

```
quant1 = 0
```

```
quant2 = 0
```

```
cont = 0
```

```
enquanto (cont < 20) {
```

```
    escreva ("\nQuantos filhos ? ")
```

```
    leia(qfilhos)
```

```
    se (qfilhos <= 2) {
```

```
        quant1 = quant1 + 1
```

```
    }
```

```
    senao {
```

```
        quant2 = quant2 + 1
```

```
    }
```

```
    cont = cont + 1
```

```
}
```

```
escreva ("\nPessoas com até 2 filhos: ", quant1)
```

```
escreva ("\nPessoas com MAIS de 2 filhos: ", quant2)
```

**Saída de
Dados**



Universidade Federal do ABC

Bacharelado em Ciência e Tecnologia Processamento da Informação

Estruturas de Repetição – Parte 1

EXERCÍCIOS

Pra você Resolver!

Exercício 01: Função $f(x)$

Escreva um algoritmo para exibir todos os valores de $f(x) = 2x^2 + 3x - 5$, para os 10 primeiros inteiros maior que 3.

- **SAÍDA:** Quais os resultados que o algoritmo dever fornecer ?
 - **ENTRADA:** Quais os insumos necessários para se obter a saída?
 - **PROCESSAMENTO:** Como transformar os insumos na saída?
-
- **SAÍDA:** $f(x)$ { para os 10 valores de x }
 - **ENTRADA:** Não há entrada! { Apenas a inicialização de $x = 4$ }
 - **PROCESSAMENTO:**
 - $Fx = 2 * (x*x) + 3*x - 5$
 - $x = x + 1$
- } 10 vezes

Exercício 02: Caixa de Supermercado

Escreva um algoritmo para simular uma caixa registradora de supermercado. O caixa deve digitar a quantidade comprada e o preço unitário de cada item. O algoritmo deve mostrar: o total da compra e a quantidade de volumes comprados

Observação:

- Não se sabe previamente quantos volumes cada pessoa tem no seu carrinho. Assim, ao digitar **quantidade comprada = 0 (zero)** indica que não há mais itens e a compra pode ser totalizada
- **SAÍDA:** total da compra e quantidade de volumes
- **ENTRADA:** quantidade e preço de cada item
- **PROCESSAMENTO:**
 - Calcular o total do item : **item = quant * precounitario**
 - Acumular o total da compra: **precototal = precototal + item**
 - Acumular o total de itens: **totalvolumes = totalvolumes + quant**

} repetir

RESUMINDO...

Coisas para não esquecer:

- Problemas que envolvam “vários elementos” podem ser tratados com **ESTRUTURAS DE REPETIÇÃO**
- A estrutura de repetição **ENQUANTO** executa o **BLOCO** de instruções “**enquanto a condição for verdadeira**”

SINTAXE DO PORTUGOL

```
enquanto <condição> {  
    INSTRUÇÃO 1  
    ...  
}
```

Coisas para não esquecer:

- **Padrão de Instrução: CONTADOR**
 - Exemplo: $\text{cont} = \text{cont} + 1$
- **Padrão de Instrução: ACUMULADOR**
 - Exemplo: $\text{soma} = \text{soma} + \text{valor}$

AGORA É
PRATICAR!!!